



RUG-III-HC

Resource Utilization Groups III
Home Care (RUG-III-HC)
**Grouping Methodology
Using iCodes**

2025



Canadian Institute
for Health Information

Institut canadien
d'information sur la santé

Production of this document is made possible by financial contributions from Health Canada and provincial and territorial governments. The views expressed herein do not necessarily represent the views of Health Canada or any provincial or territorial government.

All rights reserved.

The contents of this publication may be reproduced unaltered, in whole or in part and by any means, solely for non-commercial purposes, provided that the Canadian Institute for Health Information is properly and fully acknowledged as the copyright owner. Any reproduction or use of this publication or its contents for any commercial purpose requires the prior written authorization of the Canadian Institute for Health Information. Reproduction or use that suggests endorsement by, or affiliation with, the Canadian Institute for Health Information is prohibited.

For permission or information, please contact CIHI:

Canadian Institute for Health Information
495 Richmond Road, Suite 600
Ottawa, Ontario K2A 4H6
Phone: 613-241-7860
Fax: 613-241-8120
cihi.ca
copyright@cihi.ca

© 2024 Canadian Institute for Health Information

interRAI HC © interRAI Corporation, Washington, D.C., 1994–2022. Modified with permission for Canadian use under licence to the Canadian Institute for Health Information. Canadianized items and their descriptions © Canadian Institute for Health Information, 2024.

How to cite this document:

Canadian Institute for Health Information. *Resource Utilization Groups III Home Care (RUG-III-HC) Grouping Methodology Using iCodes*. Ottawa, ON: CIHI; 2024.

Cette publication est aussi disponible en français sous le titre *Méthodologie des groupes d'utilisation des ressources en services à domicile, version III (RUG-III-HC), avec iCodes*.

Table of contents

Introduction	4
Overview of the RUG-III-HC grouping methodology using iCodes	4
interRAI HC data source	4
Case Mix Index for RUG-III-HC groups	4
RUG-III-HC methodology: IRRS version	5
RUG-III-HC and IRRS	5
SAS code for the RUG-III-HC grouping methodology	5
Flowcharts	24
RUG-III-HC overview	24
IRRS RUG-III-HC Section 1	26
IRRS RUG-III-HC Section 2: Calculate RUG-III-HC temporary variables	31
IRRS RUG-III-HC Section 3: Calculate RUG-III-HC category triggers	61
IRRS RUG-III-HC Section 4: Assign qualifying RUG-III-HC groups	66
Appendix: Summary of product changes for RUG-III-HC using iCodes	73

Introduction

Welcome to the *Resource Utilization Groups III Home Care (RUG-III-HC) Grouping Methodology Using iCodes* prepared by the Canadian Institute for Health Information (CIHI).

The RUG-III-HC methodology is used to determine how assessments submitted to CIHI's Integrated interRAI Reporting System (IRRS) are assigned to discrete groups.

Overview of the RUG-III-HC grouping methodology using iCodes

interRAI HC data source

The foundation data for the RUG-III-HC grouping methodology using iCodes is collected using the interRAI Home Care © (interRAI HC) assessment tool developed by the International Resident Assessment Instrument Corporation (interRAI) (interrai.org).

The objective of a grouping methodology is to categorize individuals into statistically and clinically similar groups. Grouping is based on different levels of acuity and supports evidence-based system- and facility-level planning, quality improvement and case mix-adjusted resource allocation.

Case Mix Index for RUG-III-HC groups

The Case Mix Index (CMI) is a value reflecting the daily relative weight of resources used by an individual within each RUG-III-HC group compared with a base resource level (the average resource use of the client population). It should be emphasized that the CMI value is not a dollar amount; rather, it is a relative value.

CIHI does not provide CMI values for the RUG-III-HC grouping methodology at this time.

RUG-III-HC methodology: IRRS version

RUG-III-HC and IRRS

This RUG-III-HC grouping methodology using iCodes material corresponds to the IRRS version of the RUG-III-HC grouping methodology.

Assumptions made for this logic

This algorithm assumes that data conforms to CIHI's IRRS data submission standards. CIHI assumes no responsibility for unexpected results arising from the use of data that does not meet CIHI's defined formats and edit specifications for IRRS data.

As with all grouping methodology, it should be acknowledged that the RUG-III-HC algorithm is not an edit routine. If inconsistent or illogical data is processed using this algorithm, inappropriate RUG-III-HC group assignment may occur.

Items included in this product (SAS code, test data, flowcharts)

a. RUG-III-HC for iCodes SAS code (PDF file)

This document is the PDF version of the introduction, SAS program code and methodology flowcharts.

b. RUG-III-HC for iCodes SAS code (text file)

The text file contains the SAS program code in machine-readable format.

The text file corresponds exactly to the information contained in the SAS code produced below.

Questions about CIHI's version of RUG-III-HC for the interRAI HC

If you have a question about the IRRS version of the RUG-III-HC grouping methodology, please submit a query through CIHI's [online eQuery tool](#).

SAS code for the RUG-III-HC grouping methodology

The SAS code presented on the following pages is also available as a machine-readable text file.

```

/*=====*/
/* © Canadian Institute for Health Information 2025 */
/* CIHI Terms of Use */
/*=====*/
/*
/* Thank you for your interest in the CIHI product below these terms. */
/*
/* You may reproduce and use this product at no charge for its intended purposes */
/* and/or for academic or government research or study purposes so long as you: */
/* - Acknowledge CIHI's ownership of this product, and */
/* - Respect the following prohibitions. */
/*
/* You may not: */
/* - Use the product in a commercial manner, e.g., incorporating into a product */
/* for sale; */
/* - Allow others to access a version of this product that does not contain these */
/* terms of use or that has been otherwise modified without CIHI 's express, */
/* written permission; or */
/* - State or suggest CIHI uses, approves, or endorses any person, entity, service, */
/* product, cause, or opinion. */
/*
/* Note: */
/* - You indicate your understanding and acceptance of these terms when you first */
/* use the product. */
/* IF YOU CANNOT ACCEPT THESE TERMS, YOU MAY NOT USE THE PRODUCT. */
/* - CIHI requests that you exercise caution or even refrain from publishing data */
/* generated using the product if the data identifies or could identify */
/* individual */
/* - health care facilities. */
/* - These terms constitute the entire agreement relating to the use of the */
/* product. */
/* - CIHI holds the rights in and title to the product. You do not obtain its */
/* ownership or any other right in it except as provided above. */
/* - The product is provided as-is, without warranties of any kind. */
/* - You hold CIHI harmless in respect of any losses or damages arising from your */
/* use of the Work. */
/* - These terms are governed by, and to be construed in accordance with the laws */
/* of the Province of Ontario and the applicable laws of Canada, without regard */
/* to principles of conflicts of laws. */
/*
/*=====*/

```

```

/*=====*/
/*  RUG-III-HC grouping methodology using iCodes for interRAI HC          */
/*  IRRS LTCF Version 2025-2026                                           */
/*=====*/
/* Program:  RUG-III-HC grouping methodology using iCodes for interRAI HC.sas */
/*                                                    */
/* Purpose:  SAS codes for grouping interRAI HC data using interRAI iCodes */
/*                                                    */
/* Created:  March 29, 2017                                                */
/*                                                    */
/* Revised:  September 3, 2024                                           */
/*           - Modification to instrumental ADL calculation for all non-private */
/*             care settings                                              */
/*                                                    */
/*           October 20, 2022                                           */
/*           - Modification to instrumental ADL calculation for assessments */
/*             in a hospital setting                                     */
/*                                                    */
/*           May 15, 2019                                                */
/*           - Correction to mood and behaviour variable value range description */
/*           - 'CIHI_' prefix of output variables dropped                */
/*             i.e. ar3H and aNR3H                                       */
/*                                                    */
/* Notes:    Grouping methodology sas code provided by interRAI          */
/*                                                    */
/* APPLIES TO:  interRAI HC                                              */
/* DESCRIPTION:  RUG-III-HC is a resource-intensity (case-mix) measurement */
/*               system designed for use in home care programs           */
/* REFERENCES:  FOR DETAILS, SEE ARTICLE BY BJORKGREN, FRIES, SHUGARMAN   */
/*               - TESTING A RUG-III BASED CASE-MIX SYSTEM FOR HOME CARE  */
/*               Canadian J. Aging, 19 (Supp. 2):106-125, (Fall) 2000.    */
/* INTERPRETATION: See article above for basic description. This code is based */
/*               on RUG-III originally developed for use in nursing homes with */
/*               the MDS Version 2.0 assessment form. The code here crosswalks */
/*               the RUG-III-HC system, designed for the MDS-HC V2.0 instrument */
/*               to the interRAI Suite interRAI HC (Home Care). In doing so */
/*               fewer iCODE items are missing than for the MDS-HC V2.0.     */
/* INPUT VARIABLES: The interRAI HC Version 2.0 variables required by the SAS */
/*               code for interRAI RUG-III/HC classification are given below. */
/*               For each variable, the interRAI 'iCODE' item and label are */
/*               given. Specification of these 'iCODES' and links to specific */
/*               items in the interRAI HC are available in the iCODE MATRIX, */
/*               available from interRAI.                                   */
/*               */
/*               Before execution of RUG-III HC classification using the SAS */
/*               code, all 58 iCODE items must be scanned for valid values as */
/*               given in the list below. Note that this version uses      */
/*               ICD-10-CA codes.                                          */
/*=====*/

```

```

/*=====*/
/* MACRO create_sCPS_scale */
/* - code from InterRAI */
/*=====*/
%MACRO create_sCPS_scale;

/*note: if iC1 = 5 then iC2a and iD1 are missing */
/* if iC1 ne 5 then no items should be missing */

** CREATE IMPAIRMENT COUNTS **;
xcps1 = 0;
if iC1 in (1,2,3)
if iD1 in (1,2,3,4)
if iC2a = 1
** IMPAIRMENT COUNT **;
then xcps1 = xcps1 + 1;
then xcps1 = xcps1 + 1;
then xcps1 = xcps1 + 1;

xcps2 = 0;
** SEVERE IMP COUNT **;
if iC1 = 3
if iD1 in (3,4)
then xcps2 = xcps2 + 1;
then xcps2 = xcps2 + 1;

** COMPUTE CPS SCALE **;
if iC1 = 5
then sCPS = 6;
else if iC1 = 4 then do;
if iG2j in (6,8)
then sCPS = 6;
else if iG2j in (0,1,2,3,4,5)
then sCPS = 5;
end;
else if (iC1 >= 0 and iC1 < 4) then do;
if xcps1 in (2,3) then do;
if xcps2 = 2
then sCPS = 4;
else if xcps2 = 1
then sCPS = 3;
else if xcps2 = 0
then sCPS = 2;
end;
else if xcps1 = 1
then sCPS = 1;
else if xcps1 = 0
then sCPS = 0;
end;

%MEND create_sCPS_scale;

```



```

/*=====*/
/* Macro to RUN the RUG-III-HC grouping methodology using iCodes */
/*=====*/
%MACRO RUG_III_HC_grouper;

/* include the macro to create sCPS scale */
%create_sCPS_scale;

** VARIABLE VALUE CHECK **;
if (iC1 = 5 or (iE3a in (0,1,2,3) and iE3b in (0,1,2,3) and iE3c in (0,1,2,3)
and iE3d in (0,1,2,3) and iE3e in (0,1,2,3) and iE3f in (0,1,2,3)))
and (iG1aa in (0,1,2,3,4,5,6,8) or iG1ab in (0,1,2,3,4,5,6,8))
and (iG1da in (0,1,2,3,4,5,6,8) or iG1db in (0,1,2,3,4,5,6,8))
and (iG1ea in (0,1,2,3,4,5,6,8) or iG1eb in (0,1,2,3,4,5,6,8))
and iG2g in (0,1,2,3,4,5,6,8) and iG2h in (0,1,2,3,4,5,6,8)
and iG2i in (0,1,2,3,4,5,6,8) and iG2j in (0,1,2,3,4,5,6,8)
and iI1e in (0,1,2,3) and iI1f in (0,1,2,3) and iI1i in (0,1,2,3)
and iI1q in (0,1,2,3) and iI1t in (0,1,2,3)
and iJ2h in (0,1,2,3,4) and iJ2i in (0,1,2,3,4) and iJ2j in (0,1,2,3,4)
and iJ2n in (0,1,2,3,4) and iJ2q in (0,1,2,3,4) and iJ2r in (0,1,2,3,4)
and iJ6c in (0,1) and iK2a in (0,1) and iK2c in (0,1)
and iK3 in (0,1,2,3,4,5,6,7,8,9)
and iL1 in (0,1,2,3,4,5) and iL4 in (0,1)
and iL5 in (0,1) and iL7 in (0,1,2,3,4)
and iN2a in (0,1,2,3) and iN2b in (0,1,2,3) and iN2d in (0,1,2,3)
and iN2e in (0,1,2,3) and iN2f in (0,1,2,3) and iN2g in (0,1,2,3)
and iN2h in (0,1,2,3) and iN2i in (0,1,2,3) and iN2j in (0,1,2,3)
and iN2k in (0,1,2,3) and iN2n in (0,1,2,3)
and (0 <=iN3eb <=999) and (0 <=iN3fb <=999) and (0 <=iN3gb <=999)
and sCPS in (0,1,2,3,4,5,6) then do;

length x_iadls x_meal x_mmed x_phon
x_adlsum x_intake x_bedmb x_trans x_toilt x_eatng
x_th_min x_reh
x_cpal x_sept x_coma
x_spec x_clin x_ext x_ext_ct x_behav 3.;

```

```

/*-----*/
/* Step I. Initialize needed variables. */
/*-----*/

** Initialize clinical category indicators **;
x_reh = 0;
x_ext = 0;
x_spec = 0;
x_clin = 0;
x_impair = 0;
x_behav = 0;

/*-----*/
/* Step II. Calculate RUG-III/HC IADL Index. */
/*-----*/
/* Impaired Cognition, Behavior Problems, and Reduced Physical */
/* Function RUG-III/HC categories. The IADL index requires */
/* scoring conversion for 3 different ADLs and then summation. */
/* The crosswalk from the MDS-HC equates the following levels: */
/* H1aa, H1da, H1ea are changed to iG1aa, iG1da, iG1ea as follows: */
/* H1aa iG1aa */
/* 0 (independent) 0 (independent) */
/* 1 (some help) 1,2,3,4 (help, supervision, limited, extensive) */
/* 2 (full help) 5 (maximal assistance) */
/* 3 (by others) 6 (total dependence) */
/* 8 (activity did not occur) 8 (activity did not occur) */
/* For RUG-III/HC we count the number of these three at the level 'full help' */
/* or more on the RAI-HC which corresponds to iG1aa of 5 or more */
/* */
/* IADL performance items are not collected in non-private home settings. */
/* When deriving outcome measures for those assessed in hospital, */
/* performance items are substituted by capacity items. */
/* */
/* H1ab, H1db, H1eb are changed to iG1ab, iG1db, iG1eb as follows: */
/* H1ab, H1db, H1eb iG1ab, iG1db, iG1eb */
/* 0 (independent) 0 (independent) */
/* 1 (some help) 1,2,3,4 (help, supervision, limited, extensive) */
/* 2 (full help) 5 (maximal assistance) */
/* 3 (by others) 6 (total dependence) */
/*-----*/

x_iadls = .;
if iA38a in (1) /* Private home, condo, apartment, assisted living setting */
then do;

if iG1aa in (0,1,2,3,4) then x_meal = 0;
else if iG1aa in (5,6,8) then x_meal = 1;

if iG1da in (0,1,2,3,4) then x_mmed = 0;
else if iG1da in (5,6,8) then x_mmed = 1;

if iG1ea in (0,1,2,3,4) then x_phon = 0;
else if iG1ea in (5,6,8) then x_phon = 1;

end;

```

```
else
if iA38a in (2,3,4) /* hospital, residential care or other settings */
then do;
    if iGlab in (0,1,2,3,4)
        then x_meal = 0;
    else if iGlab in (5,6)
        then x_meal = 1;
    if iGldb in (0,1,2,3,4)
        then x_mmed = 0;
    else if iGldb in (5,6)
        then x_mmed = 1;
    if iGleb in (0,1,2,3,4)
        then x_phon = 0;
    else if iGleb in (5,6)
        then x_phon = 1;
end;
x_iadls = x_meal + x_mmed + x_phon;
```

```

/*-----*/
/* Step III. Calculate RUG-III ADL Index. */
/* The ADL index is required for use in splitting the Rehabilitation, */
/* Special Care, Clinically Complex, Impaired Cognition, Behavior */
/* Problems, and Reduced Physical Function RUG-III clinical */
/* categories. The ADL index requires scoring conversion for */
/* 4 different ADLs and then summation. */
/*-----*/
/* ADD calculation of ADL sum HERE */
x_adlsum = .;

** ADL scoring conversion for Bed mobility ;
IF ( iG2i = 0 OR iG2i = 1 OR iG2i = 2 ) THEN x_bedmb = 1;
ELSE IF ( iG2i = 3 ) THEN x_bedmb = 3;
ELSE IF ( iG2i = 4 ) THEN x_bedmb = 4;
ELSE IF ( iG2i = 5 OR iG2i = 6 OR iG2i = 8 ) THEN x_bedmb = 5;

** ----- ;
** ADL scoring conversion for Transfer ;
IF ( iG2g = 0 OR iG2g = 1 OR iG2g = 2 ) THEN x_trans = 1;
ELSE IF ( iG2g = 3 ) THEN x_trans = 3;
ELSE IF ( iG2g = 4 ) THEN x_trans = 4;
ELSE IF ( iG2g = 5 OR iG2g = 6 OR iG2g = 8 ) THEN x_trans = 5;

** ----- ;
** ADL scoring conversion for Toilet Use ;
IF ( iG2h = 0 OR iG2h = 1 OR iG2h = 2 ) THEN x_toilt = 1;
ELSE IF ( iG2h = 3 ) THEN x_toilt = 3;
ELSE IF ( iG2h = 4 ) THEN x_toilt = 4;
ELSE IF ( iG2h = 5 OR iG2h = 6 OR iG2h = 8 ) THEN x_toilt = 5;

** ----- ;
** Code parenteral\enteral intake level--used for Eating scoring conversion and;
** later for Special Care and Clinically Complex qualification. ;
** x_intake = 1 if (1) 51% or more of total calories are received ;
** through parenteral/enteral intake (iK5) or (2) 26% to 50% ;
** of total calories received through parenteral/enteral ;
** intake (iK5) and fluid intake is 501 or more cc per day ;
** (iK2b). ;
** NOTE: The interRAI HC does not have any measure of intake, ;
** so assume that x_intake=1 in the following code ;
** x_intake = 0 if parenteral/enteral intake is at a lower level (including ;
** none). ;

x_intake = 1;

** ADL scoring conversion for Eating ;
IF ( iK3 = 6 OR iK3 = 7 OR iK3 = 8 ) OR (iK3 = 5 AND x_intake = 1 ) ;
THEN x_eatng = 3;
ELSE IF ( iG2j = 0 OR iG2j = 1 OR iG2j = 2 ) THEN x_eatng = 1;
ELSE IF ( iG2j = 3 ) THEN x_eatng = 2;
ELSE IF ( iG2j = 4 OR iG2j = 5 OR iG2j = 6 OR iG2j = 8 ) THEN x_eatng = 3;

** ----- ;
** Sum the converted ADL scores to x_adlsum; ;
x_adlsum = x_bedmb + x_trans + x_toilt + x_eatng;

```

```

/*-----*/
/* Step IV. Determine Rehab and Rehab nursing variables needed for */
/* Rehab categories. */
/* Variables needed to determine Rehab clinical hierarchy */
/* qualification for Standard rehab classification */
/* - Total minutes of rehab therapy received (x_th_min) */
/* - Total number of days of rehab received (x_th_day) */
/* - Number of types of rehab therapy (x_th_ty3) received */
/* at 3+ days each */
/* - Number of types of rehab therapy (x_th_ty5) received */
/* at 5+ days each */
/*-----*/

** ----- ;
** Determine x_th_min -- the number of minutes of rehab therapy received. ;

x_th_min = 0;

** Add speech therapy minutes ;
IF ( iN3gb > 0 ) THEN x_th_min = x_th_min + iN3gb;

** Add occupational therapy minutes ;
IF ( iN3fb > 0 ) THEN x_th_min = x_th_min + iN3fb;

** Add physical therapy minutes ;
IF ( iN3eb > 0 ) THEN x_th_min = x_th_min + iN3eb;

if x_th_min >= 120 then x_reh= 1;
** ----- ;

/*-----*/
/* Step V. Test Extensive Care qualification. */
/*-----*/
** ;
** ----- ;
** Determine qualification for Extensive Care. ;
** ;
** Check for required Extensive Care clinical indicators. ;
** ----- ;
** Resident qualifies for Extensive Care category on the basis of clinical ;
** indicators. ;
** Qualifications ;
** Parenteral/IV feedings (iK3) OR IV medication (iN2d) OR ;
** suctioning (iN2g) OR tracheostomy care (iN2h) OR ;
** ventilator or respirator (iN2j). ;
** ----- ;

IF ( iK3 = 7 OR iK3 = 8 OR
iN2d = 2 OR iN2d = 3 OR
iN2g = 2 OR iN2g = 3 OR
iN2h = 2 OR iN2h = 3 OR
iN2j = 2 OR iN2j = 3)
THEN x_ext = 1;

```

```

/*-----*/
/* Step VI.  Test Special Care qualification.                                */
/*-----*/

x_cpal=0;

** Code Cerebral Palsy from ICD-10 CA Code                                ;
** ICD-10 CA CODES for cerebral palsy: G80.x                               ;
IF SUBSTR(iI2abb,1,3)= 'G80' THEN x_cpal=1;
IF SUBSTR(iI2bbb,1,3)= 'G80' THEN x_cpal=1;
IF SUBSTR(iI2cbb,1,3)= 'G80' THEN x_cpal=1;
IF SUBSTR(iI2dbb,1,3)= 'G80' THEN x_cpal=1;
IF SUBSTR(iI2ebb,1,3)= 'G80' THEN x_cpal=1;
IF SUBSTR(iI2fbb,1,3)= 'G80' THEN x_cpal=1;

x_sept=0;

** Code Septicemia from ICD-10 CA Code                                    ;
** CIHI is providing guidance to use A40. or A41. for coding sepsis using    ;
** ICD-10 CA however the RUG-III HC interRAI HC sas code and flow charts    ;
** documents include a broader list of ICD-10 CA sepsis codes.              ;
** ICD-10 CA CODES for septicemia: A40 A41 R65 B00.7 A02.1 A22.7 A24.1 A26.7 ;
**                                     A32.7 A42.7 B37.7                       ;
IF SUBSTR(iI2abb,1,3) in ('A40', 'A41', 'R65') THEN x_sept=1;
IF SUBSTR(iI2bbb,1,3) in ('A40', 'A41', 'R65') THEN x_sept=1;
IF SUBSTR(iI2cbb,1,3) in ('A40', 'A41', 'R65') THEN x_sept=1;
IF SUBSTR(iI2dbb,1,3) in ('A40', 'A41', 'R65') THEN x_sept=1;
IF SUBSTR(iI2ebb,1,3) in ('A40', 'A41', 'R65') THEN x_sept=1;
IF SUBSTR(iI2fbb,1,3) in ('A40', 'A41', 'R65') THEN x_sept=1;

IF SUBSTR(iI2abb,1,5) in ('B00.7', 'A02.1', 'A22.7', 'A24.1', 'A26.7',
                          'A32.7', 'A42.7', 'B37.7') THEN x_sept=1;
IF SUBSTR(iI2bbb,1,5) in ('B00.7', 'A02.1', 'A22.7', 'A24.1', 'A26.7',
                          'A32.7', 'A42.7', 'B37.7') THEN x_sept=1;
IF SUBSTR(iI2cbb,1,5) in ('B00.7', 'A02.1', 'A22.7', 'A24.1', 'A26.7',
                          'A32.7', 'A42.7', 'B37.7') THEN x_sept=1;
IF SUBSTR(iI2dbb,1,5) in ('B00.7', 'A02.1', 'A22.7', 'A24.1', 'A26.7',
                          'A32.7', 'A42.7', 'B37.7') THEN x_sept=1;
IF SUBSTR(iI2ebb,1,5) in ('B00.7', 'A02.1', 'A22.7', 'A24.1', 'A26.7',
                          'A32.7', 'A42.7', 'B37.7') THEN x_sept=1;
IF SUBSTR(iI2fbb,1,5) in ('B00.7', 'A02.1', 'A22.7', 'A24.1', 'A26.7',
                          'A32.7', 'A42.7', 'B37.7') THEN x_sept=1;

```

```

** -----
** Check for required Special Care clinical indicators
** -----
** Resident qualifies for Special Care category on the basis of
** clinical indicators.
**   Qualifications (any one sufficient)
**     1. Stage 3 or 4 pressure ulcer (iL1) (Note: Do not have
**        count of ulcers),
**        AND
**        turning and positioning (iN2n).
**     2. Feeding tube (iK3) WITH parenteral/enteral intake
**        (x_intake) AND aphasia (iJ2j).
**     3. Major skin problems (iL4) or Skin tears or cuts (iL5),
**        with wound care (iN2k)
**     4. Respiratory therapy for 7 days (iN3ia) -- MISSING ON
** interRAI HC
**     5. Cerebral palsy (x_cpal) AND ADL score of 10 or more
**        (x_adlsum).
**     6. Fever (iJ2q)
**        AND
**        vomiting (iJ2n) OR weight loss (iK2a) OR tube feeding
**        (iK3) WITH high parenteral/enteral intake (x_intake)
**        OR pneumonia (iI1q) OR dehydrated (iK2c).
**     7. Multiple sclerosis (iI1f) AND ADL score of 10 or more
**        (x_adlsum).
**     8. Quadriplegia (iI1i) AND ADL score of 10 or more
**        (x_adlsum).
**     9. Radiation therapy (iN2f).

IF (      ( iL1 = 3 OR iL1 = 4 ) AND ( iN2n = 2 OR iN2n = 3 ) )
OR ( ( iK3 = 6 OR ( iK3=5 AND x_intake = 1 ) ) AND
      ( iJ2j = 2 OR iJ2j = 3 OR iJ2j = 4 ) )
OR ( ( iL4 = 1 OR iL5 = 1 ) AND ( iN2k = 2 OR iN2k = 3 ) )

OR ( x_cpal=1 AND x_adlsum >= 10 )
OR ( ( iJ2q =2 OR iJ2q = 3 OR iJ2q = 4 )
      AND
      ( ( iJ2n =2 OR iJ2n = 3 OR iJ2n = 4 )
        OR iK2a = 1
        OR ( iK3 = 6 OR ( iK3=5 AND x_intake = 1 ) )
        OR ( iI1q = 1 OR iI1q = 2 OR iI1q = 3 OR iK2c=1 ) ) )
OR ( ( iI1f = 1 OR iI1f = 2 OR iI1f = 3 ) AND x_adlsum >= 10 )
OR ( ( iI1i = 1 OR iI1i = 2 OR iI1i = 3 ) AND x_adlsum >= 10 )
OR ( iN2f = 2 OR iN2f =3 ) )
THEN
    x_spec = 1;

```

```

/*-----*/
/* Step VII. Test Clinically Complex qualification. */
/*-----*/

** ----- ;
** Calculate x_coma indicating whether the resident is comatose with qualifiers ;
**      x_coma = 1 if resident is comatose (iC1 = 5) and not awake most of the;
**      time (iS3=1,2, or 3) and ADL dependent (iG2i, iG2g, iG2j, ;
**      and iG2h all have values of 6 or 8). ;
**      = 0 otherwise. ;

x_coma = 0;

IF ( iC1 = 5
    AND ( iG2i = 6 OR iG2i = 8)
    AND ( iG2g = 6 OR iG2g = 8)
    AND ( iG2j = 6 OR iG2j = 8)
    AND ( iG2h = 6 OR iG2h = 8) )
    THEN
        x_coma = 1;

```



```

** -----
** Check for Clinically Complex qualification.
** Resident qualifies for Clinically Complex category on the basis of
** clinical indicators.
** -----
** Qualifications (any one sufficient)
** 1. Feeding tube (iK3) WITH high parenteral/enteral
** intake (x_intake).
** 2. Comatose (iC1=5) AND not awake (iS3) AND
** ADL dependent (iG2i, iG2g, iG2j, iG2h).
** 3. Septicemia (x_sept).
** 4. Burns--second or third degree (not available separately)
** 5. Dehydration (iK2c).
** 6. Hemiplegia/hemiparesis (iI1e) and ADL score of
** 10 or more (x_adlsum).
** 7. Internal bleeding (iJ2r).
** 8. Pneumonia (iI1q).
** 9. End stage disease (iJ6c).
** 10. Chemotherapy (iN2a).
** 11. Dialysis (iN2b).
** 12. Physician order changes (iN8) on 4 or more days
** AND physician visits (iN7) on 1 or more days.
** 13. Physician order changes (iN8) on 2 or more days
** AND physician visits (iN7) on 2 or more days.
** 14. Diabetes (iI1t) AND injections (not available) on 7 days
** AND physician order changes (iN8) on 2 or more
** days. (Eventually, get insulin injections from drug list;
** 15. Transfusions (iN2i).
** 16. Oxygen therapy (iN2e).
** 17. Infection on foot (m6b) OR open lesion on foot (m6c)
** AND
** application of dressings to foot (m6f)
** Replaced by Foot problems that limit/prevent walking (iL7)
**
IF ( (iK3=6) OR (iK3 = 5 AND x_intake = 1)
OR x_coma = 1
OR x_sept = 1
OR iK2c = 1
OR ( ( iI1e = 1 OR iI1e = 2 OR iI1e = 3) AND x_adlsum >= 10)
OR ( iJ2r = 2 OR iJ2r = 3 OR iJ2r = 4)
OR ( iI1q = 1 OR iI1q = 2 OR iI1q = 3)
OR iJ6c = 1
OR ( iN2a = 2 OR iN2a = 3)
OR ( iN2b = 2 OR iN2b = 3)

OR (iI1t = 1 OR iI1t = 2 OR iI1t = 3)

OR ( iN2i = 2 OR iN2i = 3)
OR ( iN2e = 2 OR iN2e = 3)
OR ( iL7 = 2 OR iL7 = 3) )
THEN
x_clin = 1;

```

```

/*-----*/
/* Step VIII. Test Cognitive Impairment qualification. */
/*-----*/
**
** Note Use interRAI Cognitive Performance Scale (sCPS)
**
** Be sure to run sCPS code before running this program
**
** -----
** Determine Cognitive Impairment qualification.
** Qualification if Cognitive Performance Scale is 3 or more.
/* %create_sCPS_scale; */
IF sCPS >= 3 THEN x_impair = 1;
ELSE x_impair = 0;

/*-----*/
/* Step IX. Test Behavior Problems qualification. */
/*-----*/
** -----
** Check for Behavior Problems qualification.
** -----
** Behavior Problems Qualifications (any one sufficient)
** 1. Wandering occurred on 1 or more days (iE3a).
** 2. Verbally abusive behavior occurred on 1 or more
** days (iE3b).
** 3. Physically abusive behavior occurred on 1 or
** more days (iE3c).
** 4. Socially inappropriate/disruptive behavior
** occurred on 1 or more days (iE3d).
** 5. Resident resisted care on 1 or more days (iE3e).
** 6. Sexually inapprop. Behav 1 or more days (iE3f)
** 7. Hallucinations (iJ2i).
** 8. Delusions (iJ2h).

IF ( (iE3a = 2 OR iE3a = 3)
OR (iE3b = 2 OR iE3b = 3)
OR (iE3c = 2 OR iE3c = 3)
OR (iE3d = 2 OR iE3d = 3)
OR (iE3e = 2 OR iE3e = 3)
OR (iE3f = 2 OR iE3f = 3)
OR (iJ2i = 2 OR iJ2i = 3 OR iJ2i = 4)
OR (iJ2h = 2 OR iJ2h = 3 OR iJ2h = 4) )
THEN
x_behav = 1;

```

```

/*-----*/
/* Step X. Classify into RUG-III/HC Groups */
/*-----*/
** NOTE: THE ADL INDEX USED TO DERIVE THE RUG-III/HC SYSTEM BASED ON THE ;
**       RAI-HC DID NOT HAVE THE ADL SUPPORT ITEMS, AND THUS RANGED FROM ;
**       4-15 (RATHER THAN 4-18). ON ALL TERRAI SUITE INSTRUMENTS, THE ;
**       ADL INDEX (x_adlsum) INCORPORATES SUPPORT CONCEPTS AND CAN RANGE ;
**       FROM 4-18. HOWEVER, TO REMAIN CONSISTENT WITH THE DERIVATION WORK ;
**       WE CONSIDER EQUIVALENT ALL INDEX VALUES FROM 15-18. ;

** ----- ;
** Classify into Rehab Groups ** ;
** All final splits based on ADL sum (x_adlsum) and iADL index (x_iadls). ;

IF x_reh = 1 THEN DO;

    IF (11 <= x_adlsum AND x_adlsum <= 18) THEN aR3H = 'RB0';
    ELSE IF ( 4 <= x_adlsum AND x_adlsum <= 10) THEN DO;
        IF x_iadls > 1 THEN aR3H = 'RA2';
        ELSE aR3H = 'RA1';
    END;
END;

** ----- ;
** Classify into Extensive Care Groups ** ;
** ;
** To be classified as Extensive Care a resident must qualify on the ;
** basis of having Extensive Care clinical indicators (x_ext = 1) and ;
** an ADL sum of 7 or more ;
** ;
** Note that residents who have Extensive Care clinical indicators but ;
** have too low an ADL score (6 or less) are classified as Special Care ;
** rather than Extensive Care. (THIS IS DONE HERE TO REMAIN COMPATIBLE ;
** WITH THE RUG-III V5.20 SPECIFICATION - EARLIER VERSIONS MOVED THESE ;
** OBSERVATIONS TO THE CLINICALLY COMPLEX CATEGORY, AS WELL, THE COUNT ;
** USED TO SPLIT THE EXTENSIVE CARE CATEGORY IS MODIFIED FROM THE ;
** ORIGINAL.) ;
** ;
** Split into Extensive Care groups is based on a count (x_ext_ct) of other ;
** hierarchy category qualifications plus existence of ;
** parenteral/IV feeding (iK3) and IV medications (iN2d) ;

ELSE IF (x_ext = 1 AND x_adlsum >=7) THEN DO;

    x_ext_ct = x_spec + x_clin + x_impair;
    IF iK3 = 8 THEN x_ext_ct = x_ext_ct + 1;
    IF iN2d in (2,3) THEN x_ext_ct = x_ext_ct + 1;

    IF (4 <= x_ext_ct AND x_ext_ct <= 5) THEN aR3H= 'SE3';
    ELSE IF (2 <= x_ext_ct AND x_ext_ct <= 3) THEN aR3H= 'SE2';
    ELSE IF (0 <= x_ext_ct AND x_ext_ct <= 1) THEN aR3H= 'SE1';
END;

```

```

** -----
** Classify into Special Care Groups **
**
** To be classified as Special Care an assessment must satisfy one of the
** two following conditions
** 1. The resident has both Special Care clinical indicators (x_spec = 1)
** and an ADL sum of 7 or more.
** 2. The resident has both Extensive Care indicators (x_ext = 1) and
** was not already classified as Extensive care (i.e., an ADL sum of 6
** or less).
**
** Note that assessments assigned Special Care clinical indicators
** and a low ADL score (6 or less) are classified as Clinically Complex
** rather than Special Care.
**
** Split into Special Care groups is based on ADL sum, slightly different than
** RUG-III.

ELSE IF ((x_spec = 1 AND x_adlsum >= 7) OR x_ext = 1) THEN DO;

    IF (14 <= x_adlsum AND x_adlsum <= 18) THEN aR3H= 'SSB';
    ELSE IF ( 4 <= x_adlsum AND x_adlsum <= 13) THEN aR3H= 'SSA';
END;

** -----
** Classify into Clinically Complex Groups **
**
** To be classified as Clinically Complex an assessment must satisfy one of
** the two following conditions
** 1. Assigned the Clinically Complex clinical indicator (x_clin = 1),
** 2. Assigned the Special Care indicator (x_spec = 1) and was not already
** classified as Special Care (i.e., an ADL sum of 6 or less).
**
** Assignment of Clinically Complex groups is based on ADL and IADL sum.

ELSE IF (x_clin = 1 OR x_spec = 1) THEN DO;

    IF (11 <= x_adlsum AND x_adlsum <= 18) THEN aR3H= 'CC0';
    ELSE IF ( 6 <= x_adlsum AND x_adlsum <= 10) THEN aR3H= 'CB0';
    ELSE IF ( 4 <= x_adlsum AND x_adlsum <= 5) THEN DO;
        IF x_iadls >= 1 THEN aR3H= 'CA2';
        ELSE aR3H= 'CA1';
    END;
END;

```

```

** -----
** Classify into Impaired Cognition Groups **
**
** To be classified as Impaired Cognition the assessment must be assigned
** the impaired cognition indicator (x_impair = 1) and an ADL sum of 10
** or less.
**
** Assignment of Impaired Cognition groups is based on ADL and iADL sum
** (Note: nursing rehabilitation is not used to define groups)
;

ELSE IF (x_impair = 1 AND 4 <= x_adlsum AND x_adlsum <= 10) THEN DO;

    IF ( 6 <= x_adlsum AND x_adlsum <= 10) THEN aR3H= 'IB0';
    ELSE IF ( 4 <= x_adlsum AND x_adlsum <= 5) THEN DO;
        IF x_iadls >= 1 THEN aR3H= 'IA2';
        ELSE aR3H= 'IA1';
    END;
END;
;

** -----
** Classify into Behavioral Problems Groups **
**
** To be classified as Behavior Problems the resident has Behavior
** Problems indicators (x_behav = 1) and an ADL sum of 10 or less
**
** Split into Behavior Problems groups is based on ADL sum and IADL sum
** (Note: nursing rehabilitation is not used to define groups)
;

ELSE IF (x_behav = 1 AND 4 <= x_adlsum AND x_adlsum <= 10) THEN DO;

    IF ( 6 <= x_adlsum AND x_adlsum <= 10) THEN aR3H= 'BB0';
    ELSE IF ( 4 <= x_adlsum AND x_adlsum <= 5) THEN DO;
        IF x_iadls >= 1 THEN aR3H= 'BA2';
        ELSE aR3H= 'BA1';
    END;
END;
;

```

```

** -----
** Classify into Reduced Physical Functions Groups **
** Assessments are classified in the Reduced Physical Functions category
** if a previous hierarchical classification has not been made
**
** Group assignment for the Reduced Physical Functions category is based on ADL;
** and IADL sum
** (Note: nursing rehabilitation is not used to define groups)

ELSE IF (11 <= x_adlsum AND x_adlsum <= 18) THEN aR3H= 'PD0';
ELSE IF ( 9 <= x_adlsum AND x_adlsum <= 10) THEN aR3H= 'PC0';
ELSE IF ( 6 <= x_adlsum AND x_adlsum <= 8) THEN aR3H= 'PB0';
** BELOW FOR 4<=x_adlsum<=5;
ELSE DO;
    IF x_iadls >= 1 THEN aR3H= 'PA2';
    ELSE aR3H= 'PA1';
END;

** RUG-III/HC CLASSIFICATION COMPLETE

** NOW ADD NUMERICAL GROUP IDENTIFIERS **;
** NOTE THAT THESE ARE DIFFERENT THAN PREVIOUS RUG-III/HC VERSIONS, BUT NOW **;
** CAN BE SORTED TO PUT RUG-III/V2 GROUPS IN THEIR LOGICAL ORDER **;

length aNR3H 3.;
IF aR3H = 'RB0' THEN aNR3H = 111 ;
ELSE IF aR3H = 'RA2' THEN aNR3H = 121 ;
ELSE IF aR3H = 'RA1' THEN aNR3H = 122 ;
ELSE IF aR3H = 'SE3' THEN aNR3H = 210 ;
ELSE IF aR3H = 'SE2' THEN aNR3H = 220 ;
ELSE IF aR3H = 'SE1' THEN aNR3H = 230 ;
ELSE IF aR3H = 'SSB' THEN aNR3H = 310 ;
ELSE IF aR3H = 'SSA' THEN aNR3H = 320 ;
ELSE IF aR3H = 'CC0' THEN aNR3H = 411 ;
ELSE IF aR3H = 'CB0' THEN aNR3H = 421 ;
ELSE IF aR3H = 'CA2' THEN aNR3H = 431 ;
ELSE IF aR3H = 'CA1' THEN aNR3H = 432 ;
ELSE IF aR3H = 'IB0' THEN aNR3H = 510 ;
ELSE IF aR3H = 'IA2' THEN aNR3H = 521 ;
ELSE IF aR3H = 'IA1' THEN aNR3H = 522 ;
ELSE IF aR3H = 'BB0' THEN aNR3H = 610 ;
ELSE IF aR3H = 'BA2' THEN aNR3H = 621 ;
ELSE IF aR3H = 'BA1' THEN aNR3H = 622 ;
ELSE IF aR3H = 'PD0' THEN aNR3H = 710 ;
ELSE IF aR3H = 'PC0' THEN aNR3H = 720 ;
ELSE IF aR3H = 'PB0' THEN aNR3H = 730 ;
ELSE IF aR3H = 'PA2' THEN aNR3H = 741 ;
ELSE IF aR3H = 'PA1' THEN aNR3H = 742 ;
ELSE aNR3H = . ;

end;

** END OF RUG-III/HC SAS CODE **;

%MEND RUG_III_HC_grouper;

```

```

/*=====*/
/* Main macro for RUG-III-HC grouping using iCodes */
/*=====*/
/* Before running the macro, users need to update following */
/* input and output datasets: */
/* %LET datain =.; */
/* %LET dataout =.; */
/* */
/* Include following statement to run the grouping methodology: */
/* %RUN_RUG_III_HC_grouper (datain=, dataout=); */
/* */
/*=====*/

%MACRO RUN_RUG_III_HC_grouper (datain, dataout);

data &dataout.;
set &datain.;

%RUG_III_HC_grouper;

run;

title "Volume by RUG_III_HC grouper";
proc freq data= &dataout.;
table aR3H * aNR3H / missing list out= FREG_RUG_III_HC_grouper;
run;

%MEND RUN_RUG_III_HC_grouper;

/*=====*/
/* END OF PROGRAM */
/*=====*/

```

Flowcharts

RUG-III-HC overview

RUG-III-HC is a case-mix system that is made up of a grouping methodology and associated CMI values. Canadian organizations that provide community-based care to seniors can apply the RUG-III-HC grouping methodology and CMI values to their assessment data. Currently, CIHI does not publish RUG-III-HC CMI values; however, jurisdictions can develop and apply RUG-III-HC CMI values to derive measures of resource use over a specified time period for a particular client or organization. Comparative and time series trending reports can be generated using RUG-III-HC to identify peers, inform resource allocation and assist with planning.

This section outlines the approach used to identify qualifying RUG-III-HC groups for interRAI HC assessments submitted to IRRS.

Calculate RUG-III-HC temporary variables

Several temporary variables are required to derive RUG-III-HC groups. The [first part of this section](#) provides a listing of the data elements used and their valid values. The [second part of this section](#) provides further details and outlines the specific criteria to derive the RUG-III-HC temporary variables.

The temporary variables are listed below.

RUG-III-HC temporary variables

x_iadls	x_trans	x_coma
x_meal	x_toilt	x_spec
x_mmed	x_eatng	x_clin
x_phon	x_th_min	x_ext
x_adlsum	x_reh	x_ext_ct
x_intake	x_cpal	x_behav
x_bedmb	x_sept	

Calculate RUG-III-HC category triggers

There are 7 RUG-III-HC categories. Each has specific criteria that trigger whether an assessment qualifies for that category, except for the last category (Reduced Physical Functions), where all assessments qualify. The [third part of this section](#) outlines the specific criteria for the RUG-III-HC category triggers.

Assign qualifying RUG-III-HC groups

Specific criteria are used to assign a qualifying RUG-III-HC group to a given IRRS interRAI HC assessment. The [fourth part of this section](#) outlines the specific criteria for the RUG-III-HC groups. A [listing of all 23 RUG-III-HC groups and their hierarchical rank orders](#) is included in this document.

IRRS RUG-III-HC Section 1

interRAI HC variables used in RUG-III-HC

Variable name (valid codes) description

Section A — Identification Information

Location of assessment

iA38a (1, 2, 3, 4) Location of assessment

Section C — Cognition

Cognitive skills for daily decision making

iC1 (0, 1, 2, 3, 4) How well client made decisions about organizing the day

Memory recall ability

iC2a (0, 1) Short-term memory OK

Section D — Communication/Hearing patterns

Making self understood (expression)

iD1 (0, 1, 2, 3, 4) Expressing information content — however able

Section E — Mood and Behaviour

Behavioural Symptoms

iE3a	(0, 1, 2, 3)	Wandering
iE3b	(0, 1, 2, 3)	Verbally abusive behavioural symptoms
iE3c	(0, 1, 2, 3)	Physically abusive behavioural symptoms
iE3d	(0, 1, 2, 3)	Socially inappropriate/disruptive behavioural symptoms
iE3e	(0, 1, 2, 3)	Resists care
iE3f	(0, 1, 2, 3)	Inappropriate sexual behaviour

Section G — Physical Functioning

IADL (instrumental activities of daily living)

Self-performance and difficulty — Functioning during the last 7 days

iG1aA	(0, 1, 2, 3, 4, 5, 6, 8)	Meal preparation — self performance
iG1dA	(0, 1, 2, 3, 4, 5, 6, 8)	Managing medication — self performance
iG1eA	(0, 1, 2, 3, 4, 5, 6, 8)	Phone use — self performance

ADL (activities of daily living)

Self-performance — Physical functioning during the last 3 days

iG2g	(0, 1, 2, 3, 4, 5, 6, 8)	Transfer toilet
iG2h	(0, 1, 2, 3, 4, 5, 6, 8)	Toilet use
iG2i	(0, 1, 2, 3, 4, 5, 6, 8)	Mobility in bed
iG2j	(0, 1, 2, 3, 4, 5, 6, 8)	Eating

Section I — Disease Diagnoses

Disease Diagnoses

Neurological

il1e	(0, 1, 2, 3)	Hemiplegia/Hemiparesis
il1f	(0, 1, 2, 3)	Multiple sclerosis
il1i	(0, 1, 2, 3)	Quadriplegia

Infections

il1q	(0, 1, 2, 3)	Pneumonia
------	--------------	-----------

Other

il1t	(0, 1, 2, 3)	Diabetes mellitus
------	--------------	-------------------

Other Disease Diagnoses

iI2abb	CCC.CC	ICD-10-CA Code for Cerebral Palsy/Septicemia
iI2bbb	CCC.CC	ICD-10-CA Code for Cerebral Palsy/Septicemia
iI2cbb	CCC.CC	ICD-10-CA Code for Cerebral Palsy/Septicemia
iI2dbb	CCC.CC	ICD-10-CA Code for Cerebral Palsy/Septicemia
iI2ebb	CCC.CC	ICD-10-CA Code for Cerebral Palsy/Septicemia
iI2fbb	CCC.CC	ICD-10-CA Code for Cerebral Palsy/Septicemia

Section J — Health Conditions***Problem Frequency (in LAST 3 DAYS)***

iJ2h	(0, 1, 2, 3, 4)	Delusions
iJ2i	(0, 1, 2, 3, 4)	Hallucinations
iJ2j	(0, 1, 2, 3, 4)	Aphasia
iJ2n	(0, 1, 2, 3, 4)	Vomiting
iJ2q	(0, 1, 2, 3, 4)	Fever
iJ2r	(0, 1, 2, 3, 4)	GI or GU bleeding

Instability of Conditions

iJ6c	(0, 1)	End-stage disease (6 months or less to live)
------	--------	--

Section K — Oral and Nutritional Status***Nutritional issues***

iK2a	(0, 1)	Weight loss (5% or more in LAST 30 DAYS or 10% or more in LAST 180 DAYS)
iK2c	(0, 1)	Dehydrated

Mode of Nutritional Intake

iK3	(0, 1, 2, 3, 4, 5, 6, 7, 8, 9)	Mode of nutritional intake
-----	--------------------------------	----------------------------

Section L — Skin Condition

Most severe pressure ulcer

iL1 (0, 1, 2, 3, 4, 5) Most severe pressure ulcer

Major skin problems

iL4 (0, 1) e.g., lesions, 2nd- or 3rd- degree burns, healing surgical wounds

Skin tears or cuts

iL5 (0, 1) Skin tears or cuts other than surgery

Foot problem

iL7 (0, 1, 2, 3, 4) e.g., bunions, hammertoes, overlapping toes, structural problems, infection, ulcers

Section O — Treatments and procedures

Treatments and programs (*within LAST 3 DAYS*)

Treatments

iN2a	(0, 1, 2, 3)	Chemotherapy
iN2b	(0, 1, 2, 3)	Dialysis
iN2d	(0, 1, 2, 3)	IV medication
iN2e	(0, 1, 2, 3)	Oxygen therapy
iN2f	(0, 1, 2, 3)	Radiation
iN2g	(0, 1, 2, 3)	Suctioning
iN2h	(0, 1, 2, 3)	Tracheostomy care
iN2i	(0, 1, 2, 3)	Transfusions
iN2j	(0, 1, 2, 3)	Ventilator or respirator
iN2k	(0, 1, 2, 3)	Wound care

Programs

iN2n (0, 1, 2, 3) Turning/repositioning program

Therapy/Nursing services (*within LAST 7 DAYS*)

iN3eb (0000–9999)	Physical therapy — total number of minutes provided in last 7 days
iN3fb (0000–9999)	Occupational therapy — total number of minutes provided in last 7 days
iN3gb (0000–9999)	Speech–language pathology and audiology services — total number of minutes provided in last 7 days

RUG-III-HC group listing

RUG-III-HC category and name	RUG-III-HC rank	RUG-III-HC group
1. Special Rehabilitation	1	1_RB
	2	1_RA2
	3	1_RA1
2. Extensive Services	4	2_SE3
	5	2_SE2
	6	2_SE1
3. Special Care	7	3_SSB
	8	3_SSA
4. Clinically Complex	9	4_CC
	10	4_CB
	11	4_CA2
	12	4_CA1
5. Impaired Cognition	13	5_IB
	14	5_IA2
	15	5_IA1
6. Behaviour Problems	16	6_BB
	17	6_BA2
	18	6_BA1
7. Reduced Physical Functions	19	7_PD
	20	7_PC
	21	7_PB
	22	7_PA2
	23	7_PA1


IRRS RUG-III-HC Section 2: Calculate RUG-III-HC temporary variables

VARIABLE: X_MEAL

x_meal (0, 1)

A temporary value for Instrumental Activities of Daily Living (IADL) related to meal preparation.


Please note: IADL performance items are collected only in assessments completed in private home, condominium, apartment or assisted living settings. When deriving x_meal for those assessed in hospital, residential care facility or other settings, performance items are substituted for capacity items.



VARIABLE: X_MMED

x_mmed (0, 1)
A temporary value for Instrumental Activities of Daily Living (IADL) related to medication management.

Please note: IADL performance items are collected only in assessments completed in private home, condominium, apartment or assisted living settings. When deriving x_mmed for those assessed in hospital, residential care facility or other settings, performance items are substituted for capacity items.




VARIABLE: X_PHON

x_phon (0, 1)

A temporary value for Instrumental Activities of Daily Living (IADL) related to phone use.

Please note: IADL performance items are collected only in assessments completed in private home, condominium, apartment or assisted living settings. When deriving x_phon for those assessed in hospital, residential care facility or other settings, performance items are substituted for capacity items.




VARIABLE: X_IADLS


x_iadls (0, 1, 2, 3)
An indicator for impairment in Instrumental Activities of Daily Living (IADL) .


Please note: IADL performance items are collected only in assessments completed in private home, condominium, apartment or assisted living settings. When deriving x_meal, x_mmed and x_phon measures for those assessed in hospital, residential care facility or other settings, performance items are substituted for capacity items.

$$x_iadls = x_meal + x_mmed + x_phon$$

VARIABLE: X_ADLSUM (1 OF 6)


VARIABLE: X_ADLSUM (2 OF 6)




VARIABLE: X_ADLSUM (3 OF 6)

VARIABLE: X_ADLSUM (4 OF 6)

x_intake (0, 1)
Indicator for Parenteral / Enteral intake of calories and fluids.




VARIABLE: X_ADLSUM (5 OF 6)

VARIABLE: X_ADLSUM (6 OF 6)


x_adlsum (4 to 18)
The Activities Of Daily Living (ADL) score used within RUG-III.


$$x_adlsum = x_bedmb + x_trans + x_toilt + x_eatng$$

VARIABLE: SCPS (1 OF 3)


VARIABLE: SCPS (2 OF 3)

xcps2 (0, 1, 2)
Count of Severe impairment used in RUG-III CPS calculation.



VARIABLE: SCPS (3 OF 3)


VARIABLE: X_IMPAIR



VARIABLE: X_CPAL

x_cpal (0, 1)
Indicator for Cerebral palsy.

Multiple disease diagnoses can be submitted for an assessment. Check all disease diagnoses values in section il2.



VARIABLE: X_TH_MIN

x_th_min (0 or more)
The total number of minutes of physical, occupational and speech therapy in the 7 day look back period from the last assessment.

Total minutes of rehab therapy =
Total minutes of speech therapy +
Total minutes of occupational therapy +
Total minutes of physical therapy


$x_th_min = iN3gb + iN3fb + iN3eb$

VARIABLE: X_SEPT (1 OF 2)

x_sept (0, 1)
Indicator for Septicemia

Please note CIHI is providing guidance to use A40. or A41. for coding sepsis using ICD-10 CA however the RUG-III-HC sas code and flow charts documents include a broader list of ICD-10 CA sepsis codes.

Multiple disease diagnoses can be submitted for an assessment. Check all disease diagnoses values in section il2.





VARIABLE: X_SEPT (2 OF 2)

x_sept (0, 1)
Indicator for Septicemia


Please note CIHI is providing guidance to use A40. or A41. for coding sepsis using ICD-10 CA however the RUG-III-HC sas code and flow charts documents include a broader list of ICD-10 CA sepsis codes.


Multiple disease diagnoses can be submitted for an assessment. Check all disease diagnoses values in section II2.




VARIABLE: X_SPEC (1 OF 4)

VARIABLE: X_SPEC (2 OF 4)




VARIABLE: X_SPEC (3 OF 4)

VARIABLE: X_SPEC (4 OF 4)





VARIABLE: X_CLIN (1 OF 4)

x_clin (0, 1)
Number of indicators for clinically complex care.




VARIABLE: X_CLIN (2 OF 4)




VARIABLE: X_CLIN (3 OF 4)

VARIABLE: X_CLIN (4 OF 4)





VARIABLE: X_EXT

x_ext (0, 1)
Temporary indicator for extensive services items.




VARIABLE: X_EXT_CT (1 OF 2)



VARIABLE: X_EXT_CT (2 OF 2)

VARIABLE: X_BEHAV




IRRS RUG-III-HC Section 3: Calculate RUG-III-HC category triggers

RUG-III-HC TRIGGERS (1 OF 5)

x_reh (0, 1)


The Special Rehabilitation trigger is set if there was at least 120 minutes of Physical Occupational or Speech therapy provided during the look back period.



RUG-III-HC TRIGGERS (2 OF 5)

The assignment of qualifying extensive services RUG-III-HC groups to assessments includes the identification of the trigger for the extensive services category. This document does not provide a specific temporary variable for assessments that trigger the extensive services category.

The RUG-III-HC Extensive Services trigger is set if there are any RUG-III-HC Extensive Care clinical items AND a higher ADL score.



RUG-III-HC TRIGGERS (3 OF 5)


The assignment of qualifying special care RUG-III-HC groups to assessments includes the identification of the trigger for the special care category. This document does not provide a specific temporary variable for assessments that trigger the special care category.

The RUG-III-HC Special Care trigger is set

a) if there are any Extensive Care clinical indicators

OR

b) if there are any Special Care clinical indicators AND there is also a higher ADL score for the assessment




RUG-III-HC TRIGGERS (4 OF 5)

The assignment of qualifying clinically complex RUG-III-HC groups to assessments includes the identification of the trigger for the clinically complex category. This document does not provide a specific temporary variable for assessments that trigger the clinically complex category.

The Clinically Complex trigger is set


- a) if there are any Special Care clinical indicators
- OR
- b) if there are any indicators for Clinically Complex care



RUG-III HC-TRIGGERS (5 OF 5)


The assignment of qualifying impaired cognition RUG-III-HC groups to assessments includes the identification of the trigger for the impaired cognition category. This document does not provide a specific temporary variable for assessments that trigger the impaired cognition category.

The Impaired Cognition TRIGGER is set when there is a Lower ADL score and a Higher CPS score




The assignment of qualifying behaviour problems RUG-III-HC groups to assessments includes the identification of the trigger for the behaviour problems category. This document does not provide a specific temporary variable for assessments that trigger the behaviour problems category.

The Behaviour Problem trigger is set when there is a lower ADL score and any Behaviour Problem item




IRRS RUG-III-HC Section 4: Assign qualifying RUG-III-HC groups


CATEGORY: SPECIAL REHABILITATION




CATEGORY: EXTENSIVE SERVICES





CATEGORY: SPECIAL CARE




CATEGORY: CLINICALLY COMPLEX

CATEGORY: IMPAIRED COGNITION



CATEGORY: BEHAVIOUR PROBLEMS

CATEGORY: REDUCED PHYSICAL FUNCTIONS



Appendix: Summary of product changes for RUG-III-HC using iCodes

Fiscal year	Grouping methodology
2025–2026	Modifications were made to calculate iADL for assessments completed in different settings.
2023–2024	Corrections were made to calculate iADL in a hospital setting.
2022–2023	Corrections were made to the value range descriptions for the Mood and Behaviour variables.
2018–2019	Initial release



CIHI Ottawa

495 Richmond Road
Suite 600
Ottawa, Ont.
K2A 4H6
613-241-7860

CIHI Toronto

4110 Yonge Street
Suite 300
Toronto, Ont.
M2P 2B7
416-481-2002

CIHI Victoria

880 Douglas Street
Suite 600
Victoria, B.C.
V8W 2B7
250-220-4100

CIHI Montréal

1010 Sherbrooke Street West
Suite 511
Montréal, Que.
H3A 2R7
514-842-2226

cihi.ca

59195-1124

